NUCLEAR DATA, INC.
Post Office Box 451
Palatine, Illinois 60067

June, 1972

IM41-8045-00
SOFTWARE INSTRUCTION MANUAL
HARDWARE MULTIPLY/DIVIDE TEST

TABLE OF CONTENTS

SECTION I
INTRODUCTION

1-1. PROGRAM SUMMARY

1-2. The Hardware Multiply/Divide Test (ND41-8045) is a diagnostic program designed to check the ND812 hardware multiply and divide circuitry.

NOTE

This diagnostic applies to ND812 Central Processors serial number 236 and up or to earlier processor equipped with the 24 X 12 divide modification.

1-3. PROGRAM AREA

1-4. This program may reside in any Memory Field octal locations Ø2ØØ through Ø425.

1-5. STARTING ADDRESS

1-6. The starting address of this program is $Ø2ØØ_8$.

1-7. EQUIPMENT CONFIGURATION

1-8. Minimum equipment required for proper operation of this program includes:

a. ND812 Central Processor equipped with the 24 X 12 divide modification (88-0397)

b. ASR33 Teletype and interface (86-0085 and 88-0481).

SECTION II
PROGRAM DESCRIPTION

2-1. MAIN ROUTINE

2-2. The Hardware Multiply/Divide Diagnostic performs incrementing multiplication and division operation using three 12-bit values, A, B, and C for the equation:

$$\frac{(A*B) + C}{C} = A + C \text{ (or as the program defines, A remainder C)}$$

where,

A and B are initially set to 1.
C is initially set to Ø.

2-3. Value A is incremented by one from 1 to 4095. When A reaches the value 4095, B is incremented to two and A is set to one. A will again be incremented to 4095, B incremented to three and A set to one. This sequence is followed until value B equals 4095 at which time C is incremented to one and the entire sequence re-initialized. Thus:

A is incremented from one to 4095 for each increment of B
and B is incremented 4095 times for each increment of C.

2-4. Printout provided by this program is in two forms. The first states the values used when an erroneous answer was computed.

A X B = Z + C/B = X R Y

where,

A, B, C, X, and Y are 12-bit numbers
Z is the 24-bit product.
R indicates remainder (X remainder Y)
and
X should = A
Y should = B

2-5. The second printout will be automatically provided each time value C is incremented (approximately every 15 minutes) and contains the accumulation errors. Printout is in the following form:

00000071 E

This feature is of value when overnight testing is desired.

2-6. Another feature is included that prevents the program from incrementing any values when an error condition is encountered. This allows troubleshooting of a particular set of values if a pattern is detected. Setting the ND812 SWITCH REGISTER Bit Ø to "1" enables the feature.

SECTION III
OPERATIONAL PROCEDURE

3-1. LOADING AND INITIALIZATION PROCEDURE

3-2. The following is a step-by-step procedure describing the program loading sequence:

a. Load the Hardware Multiply/Divide Test (ND41-8045) into any Memory Field with the Binary Loader or Hardware Loader. Refer to IM41-0005 for loading procedure.

b. Set the ND812 SWITCH REGISTER to $\emptyset 2\emptyset\emptyset_8$ and depress LOAD AR key.

c. Depress the ND812 START key.

d. The program will start and continue to operate until the ND812 STOP key is depressed.

SECTION IV
OPERATOR OR USER CONTROL

4-1. GENERAL INFORMATION

4-2. In addition to the initiation and termination procedure outlined in Section III, the value redundancy control is the only operator control. Value redundancy is accomplished by setting the ND812 SWITCH REGISTER Bit Ø to "1" and causes the program to continually execute the diagnostic test using the values used when an error was detected. To recover from the redundancy operation, set the ND812 SWITCH REGISTER Bit Ø to "Ø".

SECTION V
ERROR DIAGNOSTICS

5-1. ERROR INDICATIONS

5-2. Detection of an error causes the program to print:

A X B = Z + C/B = X R Y

where,

A, B, C, X and Y are 12-bit numbers
Z is the 24-bit product
R indicates remainder (X remainderY)
and
X should = A
Y should = B

5-3. An accumulative error message is printed for every incrementation of value C and is in the following form:

00000071 E

SECTION VI
COMMAND SUMMARY

6-1. GENERAL

6-2. This program does not use keyboard entry command. The only controls are the ND812 STOP key for termination and SWITCH REGISTER Bit Ø for the value redundancy operation.

SECTION VII
FLOW CHARTS

7-1. GENERAL

7-2. Attached pages 7-2 and 7-3 is a flow chart of the Hardware Multiply/Divide Test.



Figure 7-1. Hardware Multiply/Divide Flow Chart (Sheet 1 of 2)



Figure 7-1. Hardware Multiply/Divide Flow Chart (Sheet 2 of 2)

SECTION VIII
PROGRAM LISTING

8-1. GENERAL

8-2. Attached is a copy of the Hardware Multiply/Divide binary listing.

```
/ND41-8045-00 HARDWARE MULTIPLY-DIVIDE TEST (24X12)

/BIT 0 LOCKS SYSTEM IN SCOPE LOOP AFTER FIRST ERROR

            *200

0200  1510  START,  CLR      J        / SET J = 0
0201 >5476           STJ      MAXREM   / SET MAXIMUM LEGAL REMAINDER = ZERO
0202  1514  LOOP3,  CLR INC  J        / SET J = 1
0203 >5475           STJ      A        / A= 1
0204 >5475           STJ      B        / B= 1
0205  1510  LOOP2,  CLR      J        / SET J = 0
0206 >5474           STJ      REM      / SET REMAINDER = ZERO
0207  6434  LOOP1,  JPS      MDTEST   / EXECUTE TEST SUBROUTINE
0210  5072           LDJ      REM      / CHECK FOR REM AT MAXREM LIMIT
0211  2466           SMJ      MAXREM   / SKIP IF DIFFERENT
0212  6006           JMP      LIMIT    / REM LIMIT REACHED - GO TO LARGER LOOP
0213  1504           INC      J        / UPDATE REM
0214  5466           STJ      REM      / SAVE IN REM
0215  2464           SMJ      B        / CHECK FOR REM AT B LIMIT
0216  6002           JMP      LIMIT    / REM LIMIT REACHED - GO TO LARGER LOOP
0217  6110           JMP      LOOP1    / NO LIMIT REACHED - REPEAT SMALLEST LOC

0220  3460  LIMIT,  ISZ      A        / A+1 INTO A, OVERFLOW?
0221  6114           JMP      LOOP2    / NO, REPEAT MULTIPLY DIVIDE LOOP
0222  3456           ISZ      A        / YES, SET A = 1
0223  3456           ISZ      B        / NO, B+1 INTO B, OVERFLOW?
0224  6117           JMP      LOOP2    / NO, CONTINUE INTERMEDIATE LOOP
0225  3452           ISZ      MAXREM   / YES, INCREMENT MAXIMUM LEGAL REMAINDER
0226  1530           SET      J        / SET J = 7777
0227  5451           STJ      A        / SET A = 7777
0230  5451           STJ      B        / SET B = 7777
0231  2301           SUBL     1        / SET J = 7776
0232  5450           STJ      REM      / SET REM = 7776
0233  6410           JPS      MDTEST
0234  7055           XCT      X2       / OUTPUT CR, LF,
0235  5051           LDJ      CNTR+1   / SET J = MSB OF COUNTER
0236  7063           XCT      X3       / OUTPUT VALUE
0237  5046           LDJ      CNTR     / SET J = TO LSB OF COUNTER
0240  7053           XCT      X1       / OUTPUT VALUE
0241  0305           305               / OUTPUT E FOR ERRORS
0242  6140           JMP      LOOP3    / DO OUTER LOOP AGAIN

0243  0000  MDTEST,  0                 / GENERAL MULTIPLY/DIVIDE SUBROUTINE
0244  1010  SCOPE,   LJSW             / LOAD J FROM SWITCH REGISTER
0245  1506           SIN      J        / IS BIT 0 ON?
0246  1410           CLR      FLAG     / NO, SET FLAG = 0
0247  5031           LDJ      A        / YES, SET J = A
0250  0510           TWLDK             / SET K = B
0251  0301           B
```

```
0252  1000          MPY             / A.B = . THEN A.B/B
0253  5026          LDJ     B       / SET J = B
0254  1303          EXJRKS          / J=LSB K=MSB R=1 S= ALTERED VALUE
0255  0550          TWSTK           / STORE RESULT OF MULTIPLY AT KV AND
0256  0374          KV              / MSB AT KV
0257  5424          STJ     JV      / LSB AT JV
0260  1450          CLR     0       / SET UP TO ADD REM
0261  4421          ADJ     REM
0262  1455          SIZ CLR 0
0263  1604          INC     K
0264  1001          DIV             / DIVIDE K,J/R, QUOTIENT IN J REM. IN
0265  1405          SIZ     FLAG    / IS FLAG = 0
0266  6122  X4,     JMP     SCOPE   / NO, GO CONTINUE IN SCOPE LOOP
0267  5415          STJ     AJ      / STORE QUOTIENT IN AJ
0270  0250          TWSMK           / IS REMAINDER CORRECT?
0271  0302          REM
0272  1442          SKIP
0273  6014          JMP     ERR     / NO, ERROR, REMAINDER SHOULD = REM
0274  2404          SMJ     A       / YES, ARE A AND QUOTIENT =?
0275  6332          JMP@    MDTEST  / ALL RESULTS O.K. - EXIT
0276  6011          JMP     ERR     / NO, ERROR THEY SHOULD BE =.

0277  0000  MAXREM,  0               / MAXIMUM ALLOWABLE REMAINDER
0300  0000  A,       0               / MULTIPLICAND
0301  0000  B,       0               / MULTIPLIER AND DIVISOR
0302  0000  REM,     0               / REMAINDER
0303  0000  JV,      0               / PRODUCT OF MULTIPLICATION - LSB
0304  0000  AJ,      0               / QUOTIENT OF DIVIDE A.B/B
0305  0000  CNTR,    0               / LSB OF COUNTER
0306  0000          0               / MSB OF COUNTER

0307  1405  ERR,    SIZ     FLAG    / SKIP IF FLAG NOT SET
0310  6033          JMP     COUNT2  / NO, GO TO COUNT2
0311 >6474  X2,     JPS     CRLF    / OUTPUT CR, LF.
0312  5112          LDJ     A
0313  6440  X1,     JPS     OCTS    / OUTPUT A
0314  0330          330
0315  5114          LDJ     B
0316  6435          JPS     OCTS    / OUTPUT B
0317  0275          275
0320  5054          LDJ     KV
0321  6444  X3,     JPS     OCT     / OUTPUT KV
0322  5117          LDJ     JV
0323  6430          JPS     OCTS    / OUTPUT JV
0324  0253          253
0325  5123          LDJ     REM
0326  6425          JPS     OCTS
0327  0257          257
0330  5127          LDJ     B
0331  6422          JPS     OCTS
```

```
0332  0275            275
0333  5127            LDJ     AJ
0334  6417            JPS     OCTS    /OUTPUT AJ
0335  0322            322
0336  1374            ROTD    JK 14
0337  6426            JPS     OCT     / OUTPUT REMAINDER
0340  1010            LJSW            / LOAD SWITCH REGISTER
0341  1502            SIP     J
0342  1420            CMP     FLAG
0343  3536  COUNT2,    ISZ     CNTR    / INCREMENT COUNTER , OVERFLOW?
0344  1442            SKIP            / NO
0345  3537            ISZ     CNTR+1  / YES, INCREMENT MSB OF COUNTER
0346  1010            LJSW            / LOAD J FROM SWITCH REGISTER
0347  1505            SIN     J       / IS BIT 0 ON?
0350  6255            JMP@    ALOOPP
0351  1430            SET     FLAG    / YES, SET FLAG = 1
0352  7164            XCT     X4      /JMP SCOPE / LOCK INTO SCOPE LOOP

0353  0000  OCTS,     0
0354  6411            JPS     OCT
0355  5045            LDJ     K240    / SET J = 240
0356  6435            JPS     TYPE    / PRINT SPACE
0357  5304            LDJ@    OCTS
0360  6433            JPS     TYPE
0361  5041            LDJ     K240
0362  6431            JPS     TYPE
0363  3510            ISZ     OCTS
0364  6311            JMP@    OCTS    / EXIT

0365  0000  OCT,      0
0366  5435            STJ     TEMP    / STORE J IN TEMP
0367  6406            JPS     OUT     / GO TO OUT
0370  6405            JPS     OUT     / "
0371  6404            JPS     OUT     / "
0372  6403            JPS     OUT     / "
0373  6306            JMP@    OCT     / EXIT

0374  0000  KV,       0               / PRODUCT OF MULTIPLICATION = MSB

0375  0000  OUT,      0
0376  5025            LDJ     TEMP    / SET J = TEMP
0377  1163            ROTD    J 3     / SHIFT MSB OF DIGIT INTO LSB
0400  5423            STJ     TEMP
0401  2107            ANDL    7       / STRIP OFF BITS 0-8
0402  4422            ADJ     K260    / ADD ASCII 0
0403  6410            JPS     TYPE    / OUTPUT DIGIT
0404  6307            JMP@    OUT     / EXIT

0405  0000  CRLF,     0
0406  5012            LDJ     K215
```

```
0407   6404           JPS      TYPE     / OUTPUT CARRIAGE RETURN
0410   5011           LDJ      K212
0411   6402           JPS      TYPE     / OUTPUT LINE FEED
0412   6305           JMP@     CRLF     / EXIT

0413   0000   TYPE,   0
0414   7413           TCP                / CLEAR PRINT FLAG, LOAD FROM J
0415   7414           TOS                / IS FLAG = 1?
0416   6101           JMP      .-1       / NO, TRY AGAIN
0417   6304           JMP@     TYPE      / YES, EXIT

0420   0215   K215,   215
0421   0212   K212,   212
0422   0240   K240,   240
0423   0000   TEMP,   0
0424   0260   K260,   260
0425   0210   ALOOPP, LOOP1+1

/E3645
```

```
SE 1310
A         = 0300
AJ        = 0304
ALOOPP    = 0425
B         = 0301
CNTR      = 0305
COUNT2    = 0343
CRLF      = 0405
ERR       = 0307
JV        = 0303
K212      = 0421
K215      = 0420
K240      = 0422
K260      = 0424
KV        = 0374
LIMIT     = 0220
LOOP1     = 0207
LOOP2     = 0205
LOOP3     = 0202
MAXREM    = 0277
MDTEST    = 0243
OCT       = 0365
OCTS      = 0353
OUT       = 0375
REM       = 0302
SCOPE     = 0244
START     = 0200
TEMP      = 0423
TYPE      = 0413
X1        = 0313
X2        = 0311
X3        = 0321
X4        = 0266
ER 0000
```